FOOD INTAKE MONITOR

Hardware and Software User's Manual
DOC-047
Rev 3.0

MED Associates, Inc.
P.O. Box 319
St. Albans, Vermont 05478
www.med-associates.com

TABLE OF CONTENTS

CHAPTER 1
Introduction

Please read this manual carefully prior to unpacking the Food Intake Monitoring System. This documentation contains important information regarding the set-up and operation of the Weighing System. Refer to the Appendix at the end of this manual for instructions on driver and software installation. Follow the instructions in the Appendix before proceeding through the remainder of this manual.

The ENV-200FCW Food Intake Monitoring System is housed on a standard modular panel containing a food hopper and a precision sensor with electronics to measure the weight of the hopper and food. The sensor output is amplified and signal conditioned for a 0 to +10 volt output range which can be input to an A/D converter, chart recorder, etc. for monitoring the weight of the food.

The 0 to 10V full-scale sensor output is accurate to 1/10 of a gram and is input to one of 16 channels of the DIG-745ADC 16 Channel PCI and Transducer Interface Package. A photo sensor detects head entry into the hopper, indicating that the animal may be eating. The weighing system is powered from an internal +/- 15-volt power supply within the interface cabinet.

The DIG-745ADC 16 Channel PCI and Transducer Interface Package consists of a DIG-744 PCI Analog-to-Digital Converter card which may be installed in the same computer with a MED-PC® interface card, an SG-244 cable that connects this card to the transducer interface module and a DIG-745 16 Channel Transducer Interface Module. This interface module may be installed in the interface cabinet (space permitting) or in a separate SG-6080C Table Top Cabinet. Each individual 16-transducer channel has a tare control used to "zero" the transducers. The DIG-745 ADC interface card supplies the amplifier with power and tare control signal.

Figure 1.1 - DIG-745ADC Package

CHAPTER 2
System Assembly

Carefully open each shipping box and remove the packing material to expose the Food Monitors. Identify and label each monitor as #1, #2, etc. as it's unpacked and identify the hopper platform, this is where the food hopper sits, refer to Figure 2.1.

NOTE: The food hopper platform is attached to the weighing sensor and must be handled with extreme care to avoid damage to the sensor.

Care must be taken to avoid damage to the sensitive load-cell sensor assemblies when moving chambers or individual modular panels containing the Food Monitor assemblies (herein referred to as 'assemblies').

Place each chamber on a sturdy surface, away from any external sources of vibration. Use caution when routing cables from the assemblies to the interface cabinet to avoid a 'rats-nest' of cables.

NOTE: Use care when placing or removing the food hopper onto the weighing platform. Never use excessive force when seating or removing the hopper. If downward pressure must be applied to the hopper/platform, support the platform from underneath while using the other hand to place the hopper. Placing excessive force or weight on the weighing platform will cause damage.

Figure 2.1 - Complete Assembly



DIG-745 Switch Setting

The DIG-745 card has 16 channels that can be used to interface both food and liquid intake monitors. There is a switch that corresponds to each channel that must be set to Liquid or Food. The default positions for the switches are as follows, switches 1 – 8 are set to Liquid and switches 9 – 16 are set to Food. This means that if more than eight of either the food or liquid intake devices are going to be connected to the DIG-745, the switch settings will need to be changed accordingly. The switches can be accessed by removing the DIG-745 card from the interface cabinet. The default switch settings require the connections shown in Figure 2.3.

It is also important to note that if the default switch settings are changed and the system is being used in conjunction with MED-PC, the MedState Notation must be changed accordingly.

Figure 2.2 - DIG-745 Channel 1 Switch



Figure 2.3 - Default DIG-745 Connections



Cable Connections

When making cable connections, first identify each type of cable, and then make the connections. Use care when connecting the cables to the mating connectors since most connectors are keyed, so proper alignment is important to avoid damage to connectors.

Amplifier Cabling

Once all the food intake monitor assemblies are placed in the chambers and the interface cabinet is situated, begin by connecting each channel of the DIG-745 card to a food monitor using the supplied 8-pin modular cables.

Insert one end of the cable into the appropriate channel connector until a 'snap' or 'click' is heard. The cable should be locked in place into the connector. Be sure to connect the food intake monitors into Channels 9 – 16 if the DIG-745 switches are in the default positions. Refer to Figure 2.3.

Connect the other end of the modular cable to a food intake monitor amplifier. Again, the cable connector is keyed and a 'click' will be heard as the cable locks in place. To release the cable from the connector, gently press down on the tab of the modular cable and remove. Figure 2.4.

NOTE: The interface cable MUST be connected to the weighing system amplifier in order for the head entry detector to function.

Figure 2.4 - Amplifier Connection

National Instruments Cabling

Connect one end of the SG-244 Cable to the **ADC** connector on the DIG-745 card connector and the other end to the DIG-744 card (in the computer). Use care when aligning and inserting this cable into the connectors as a misalignment of the connector could cause the pins to fold over and bend. Secure the cable to the connector by tightening the jackscrews located on both sides of the cable connector.

Smart Control Connection Panel Cabling

Refer to the manual "Wiring the Operant Test Chamber Package" and the supplied wiring chart for instructions on how to wire the Operant Chamber and Interface Cabinet.

Set-Up Procedure Using the Upper Food Hopper

1. Remove the hopper from the platform by first removing the calibration weight support and loosening the hopper security thumbscrew. Carefully lift and remove the hopper from the platform, then remove the hopper cover.
2. Fill the hopper with up to 200 grams of solid food and replace the cover to prevent spillage out the top as the subject aggressively feeds.
3. Gently return the hopper to the weighing platform and tighten the thumbscrew. Verify that the hopper is not in contact with any other parts of the assembly, i.e. cabling, IR sensor supports, etc. Place the calibration weight support back onto the assembly. See Figure 2.5.

Figure 2.5 - Upper Food Hopper Assembly



Figure 2.6 shows two types of upper food hoppers. One with the optional custom "Crysco" style food receptacle (ENV-200FCW-C) and the other with the standard food spillage catcher.

Figure 2.6 - Upper Food Hoppers (ENV-200FCW-C on left and ENV-200FCW on right)

For detailed instructions on switching between the ENV-200FCW and the ENV-200FCW-C, refer to the "ENV-200FCW-C to ENV-200FCW Conversion Manual".

CHAPTER 3
Weighing System Tare and Calibrate Utility

Weighing System Main Screen

To run the Weighing System Tare and Calibrate Utility, open the **Weighing System** application. The screen shown in Figure 3.1 will appear.

Figure 3.1 - Weighing System Tare and Calibration Utility



Each channel has it's own display and Reset/Tare button. Click the **Reset** button for each of the channels being used. This will turn the displays from red to white. If any of the displays remain red, an error has occurred in that particular channel's reading of the digital Tare potentiometers. An error can occur if the channel is not connected to a device or if the interface cabinet is not turned on before the Weighing System Tare and Calibration Utility was started.

If the channel on which the error occurred is not being used for the experiment, the error can be ignored. If the channel is in use and was not connected, connect the device to the interface and click on the corresponding **Reset** button. If the interface cabinet was not turned on when the utility was started, exit the program. Restart the program after power has been applied to the interface cabinet.

Before beginning the calibration procedure, be sure to select the correct system in the **Weighing System** area of the screen shown in Figure 3.1.

Help Screen

Clicking on **Help** will cause the screen shown in Figure 3.2 to appear. Clicking on any of the tabs on this screen will display detailed information about each function of the software.

Figure 3.2 - Help Screen

Calibrating the Channels

1. Attach the empty hopper to the Food Intake Monitor as shown in Figure 2.1.
2. Click the **Tare** buttons associated with each channel being used. After taring, the displays for all the channels that have devices connected to them should read **0.0**. If they do not read zero, double-click on the display for that channel. This will cause the **Reset** button to change back to **Tare** so that the channel may be tared again. If it still does not read zero, contact Med Associates for assistance.
3. Minimize the **Weighing System Tare and Calibrate Utility** and open the **National Instruments Measurement and Automation Explorer** Software. The screen shown in Figure 3.3 will appear.

Figure 3.3 - Measurement and Automation Explorer



4. Expand the **Devices and Interfaces** folder to view the **PCI-6023E (Device 1)** interface card. The screen shown in Figure 3.4 will appear.

Figure 3.4 - Devices and Interfaces



5. Click the **Properties** button from the top tool bar, and then click on the **AI** tab. Within the **AI** tab, select from the **Mode** menu **Referenced Single Ended**, for initial setup only. Click **OK** to exit.

6. Click the **Test Panel** button from the top tool bar. The screen shown in Figure 3.5 will appear.

Figure 3.5 - Test Panel



7. Ensure the **Data Mode** is set to **Strip Chart** and the **Y Scale Mode** is **Full Range**, as shown in Figure 3.5.
8. Change the Test Panel data channel to match the intake monitor that is being calibrated. To calibrate the first monitor, set the channel to 0, for the second monitor, set the channel to 1, etc.
9. Verify that the position of the green signal line in the chart and the **Average Reading** below the chart is approximately zero. Now place the calibration weight on the platform. If using the ENV-200FCW, this weight should be 200 grams. If using the ENV-200FCW-C this weight should be 100 grams.
10. Using a small flat blade screwdriver, turn the Gain Adjustment Screw located through the top cover of the amplifier box until the **Average Reading** is +9.5 ± 0.1 Volts (+9.40 to +9.6 Volts). See Figure 3.6.

Figure 3.6 - Gain Adjustment Screw

11. Maximize the **Weighing System Tare and Calibrate Utility**. Enter the value of the calibration weight being used in the **Calibration Weight** box. If using the ENV-200FCW, the calibration weight should be 200 grams. If using the ENV-200FCW-C the calibration weight should be 100 grams. This will be the new full-scale weight. Select the **CAL** button.
12. The **Current Value** box should now contain the calibration weight value. When the calibration weight is removed, the **Current Value** should go down to 0.0 grams. If the **Current Value** does not show 0.0 grams, re-tare the appropriate channels to 0.0 grams.
13. Once again, place the calibration weights back onto the hoppers, and with a **Current Value** reading equal to the calibration weight value, click the **Reset** buttons. Now when the weights are removed, the **Current Value** should be 0.0 grams. The calibration information for all channels will be saved when the software is exited.

CHAPTER 4
Sample MED-PC Programs

Food Weighing System.mpc

When the program starts, it looks for a Head Entry into the Food Hopper. When a Head Entry is detected the program times how long the meal is and records how much food is consumed during the meal. The "Minimum Meal Weight" parameter is used to determine if the eating bout is a Partial Meal or a Complete Meal. The "End of Meal Criterion" is used to determine when a meal has ended. The default value is two minutes. This means that two minutes must pass with no Head Entry into the Food Hopper before the meal is considered over. It is not used to determine if the meal was a Partial Meal or a Complete Meal. The "Session Time" is how long the program will run. This program expects a Resolution of 1ms.

Compare the configurations below with the wiring chart supplied with all system orders. Edit input and output numbers if they are different for the system, or contact MED Associates for assistance.

Standard Input Configuration:

Head Entry = 1 (Level Mode)

Standard Output Configuration:

House Light = 7

The sample program is as follows:

```
\ Copyright (C) 2008 MED Associates, All rights reserved.

\ Food Weighing System.mpc
\
\ After the program starts it looks for a Head Entry into the Food Hopper.  When
\ a Head Entry is detected the program times how long the meal is and records
\ how much food is eaten during the Meal.
\
\ The "Minimum Meal Weight" parameter is used to determine if the Eating Bout is
\ a Partial Meal or a Complete Meal.
\
\ The "End of Meal Criterion" is used to determine when a Meal has ended.  The
\ default value is two minutes.  This means that two minutes must pass with no
\ Head Entry into the Food Hopper before the Meal is considered over.  It is not
\ used to determine if the Meal was a Partial Meal or a Complete Meal.
\
\ The "Session Time" is how long the program will run.
\
\ Program expects a Resolution of 1ms.


\ Inputs
^HeadEntry  = 1  \ Level Mode

\ Outputs
^HouseLight = 7


\ A() = Control Variables with Assigned Aliases as Defined
Var_Alias Minimum Meal Weight (grams)     = A(0)  \ Default = 0.1 grams
Var_Alias End of Meal Criterion (seconds) = A(1)  \ Default = 120 seconds
Var_Alias Session Time (minutes)          = A(2)  \ Default = 480 minutes
```

```
^MinMeal      = 0
^EndMeal      = 1
^Session      = 2
^EndMealTicks = 3


\ List Data Variables Here
\  E = Eaten Food Total (grams)
\  F = Starting Food Total (Starting Weight) (grams)

\  M() = Meal Data
\  M(0)   = Total Number of Meals
\  M(1)   = Total Food Consumed in Meals (grams)
\  M(I)   = Length of Meal (ms)
\  M(I+1) = Food Consumed During Meal (grams)

\  P() = Partial Meal Data
\  P(0)   = Total Number of Partial Meals
\  P(1)   = Total Food Consumed in Partial Meals (grams)
\  P(J)   = Length of Partial Meal (ms)
\  P(J+1) = Food Consumed During Partial Meal (grams)

\  W = Ending Food Total (Ending Weight) (grams)


\ List Working Variables Here
\  B = Food Eaten this Bout(grams)
\  C = Current Food Available (grams)
\  I = Subscript for the Meal Array M
\  J = Subscript for the Partial Meal Array P
\  S = Meal Timer
\  T = Session Timer


\ Z-Pulses Used in this Program
^StartMeal = 1   \ Signal Start of a Meal
^StopMeal  = 2   \ Signal Stop of a Meal
^End       = 32  \ Signal End of Data Collection


DIM A = 3
DIM M = 10000
DIM P = 10000


DISKFORMAT  = 8.3
DISKCOLUMNS = 2


\************************************************
\           FOOD WEIGHING SYSTEM SCHEDULE
\  S1 - Set Default Values
\       Minimum Meal Weight          (0.1 g)
\       End of Meal Criterion        (120 sec)
\       Session Time                 (480 min)
\************************************************
S.S.1,
S1,
  0.01": SET A(^MinMeal) = 0.1, A(^EndMeal) = 120, A(^Session) = 480;
         SET I = 2, J = 2, M(I) = -987.987, P(J) = -987.987 ---> S2

S2,
  #START: CLEAR 1,60;
          SET A(^EndMealTicks) = A(^EndMeal) * 1" ---> S3
  0.01": SHOW 2,Minimum Meal Weight (grams),A(^MinMeal);
         SHOW 3,End of Meal Criterion (sec),A(^EndMeal);
         SHOW 4,Session Time (min),A(^Session) ---> SX

S3,
  1': SET A(^EndMealTicks) = A(^EndMeal) * 1" ---> SX
```

```
\*************************************************
\                    RECORD MEALS
\*************************************************
S.S.2,
S1,
  #START: ~F := Weigh(MG, BOX);~;
          SHOW 6,Starting Food Weight,F;
          ON ^HouseLight ---> S2

S2,
  #R^HeadEntry: Z^StartMeal ---> S3
  #Z^End: ~W := Weigh(MG, BOX);~ ---> S4

S3,
  #R^HeadEntry: ---> S3
  A(^EndMealTicks)#T: ~W := Weigh(MG, BOX);~; SET B = C - W; Z^StopMeal;
                      IF B >= A(^MinMeal) [@Meal, @NoMeal]
                         @Meal: ADD M(0); SET M(I) = S, M(I+1) = B, M(I+2) = -987.987;
                                SET I = I + 2, M(1) = M(1) + B, E = E + B, C = W ---> S2
                         @NoMeal: ADD P(0); SET P(J) = S, P(J+1) = B, P(J+2) = -987.987;
                                  SET J = J + 2, P(1) = P(1) + B, E = E + B, C = W ---> S2
  #Z^End: ~W := Weigh(MG, BOX);~; SET B = C - W;
          IF B >= A(^MinMeal) [@Meal, @NoMeal]
             @Meal: ADD M(0); SET M(I) = S, M(I+1) = B, M(I+2) = -987.987;
                    SET I = I + 2, M(1) = M(1) + B, E = E + B, C = W ---> S4
             @NoMeal: ADD P(0); SET P(J) = S, P(J+1) = B, P(J+2) = -987.987;
                      SET J = J + 2, P(1) = P(1) + B, E = E + B, C = W ---> S4

S4,
  1': ---> SX


\*************************************************
\                     MEAL TIMER
\*************************************************
S.S.3,
S1,
  #Z^StartMeal: SET S = 0.001 ---> S2

S2,
  0.001": SET S = S + 0.001 ---> S2
  #Z^StopMeal ! #Z^End: ---> S1


\*************************************************
\                   UPDATE DISPLAY
\*************************************************
S.S.4,
S1,
  #START: SHOW 2,Total # of Meals,M(0);
          SHOW 3,Total Food Eaten in Meals,M(1);
          SHOW 4,Total # of Partial Meals,P(0);
          SHOW 5,Total Food Eaten in Partial Meals,P(1) ---> S2

S2,
  1": SHOW 2,Total # of Meals,M(0);
      SHOW 3,Total Food Eaten in Meals,M(1);
      SHOW 4,Total # of Partial Meals,P(0);
      SHOW 5,Total Food Eaten in Partial Meals,P(1) ---> S2


\*************************************************
\                    TIME SESSION
\*************************************************
S.S.5,
S1,
  #START: ---> S2

S2,
  1': ADD T; SHOW 1,Session Time (min),T;
      IF T >= A(^Session) [@End, @Continue]
         @End: OFF ^HouseLight; Z^End ---> S3
         @Continue: ---> S2

S3,     \ Allow time for screen update before closing the session and saving data.
  5": ---> STOPABORTFLUSH
```

```
\************************************************
\                   UPDATE WEIGHT
\************************************************
S.S.6,  \ Get the Starting Weight.
        \
        \ Update the Starting Weight every 1 minute
        \ when the Animal is not eating.  This removes
        \ food that is lost for reasons other than the
        \ Animal eating.
S1,
  0.01": ~C := Weigh(MG, BOX);~; ---> S2

S2,
  #R^HeadEntry: ---> S3
  1': ~C := Weigh(MG, BOX);~; ---> S2
  #Z^End: ---> S4

S3,
  #R^HeadEntry: ---> S3
  A(^EndMealTicks)#T: ---> S2
  #Z^End: ---> S4

S4,
  1': ---> SX
```

Liquid & Food Weighing System.mpc

After the program starts it looks for a Head Entry into the Food and/or Liquid Hopper. When a Head Entry is detected the program times how long the meal and/or drinking session is and records how much food and/or liquid is taken during the session. The "Minimum Meal Weight" parameter is used to determine if the eating bout is a Partial Meal or a Complete Meal. The "End of Meal Criterion" is used to determine when a meal has ended. The default value is two minutes. This means that two minutes must pass with no Head Entry into the Food Hopper before the meal is considered over. It is not used to determine if the meal was a Partial Meal or a Complete Meal.

The "Minimum Drink Weight" parameter is used to determine if the drinking session is a Partial Drinking Bout or a Complete Drinking Bout. The "End of Drinking Bout Criterion" is used to determine when a drinking session has ended. The default value is two minutes. This means that two minutes must pass with no Head Entry into the Liquid Hopper before the drinking session is considered over. It is not used to determine if the drinking session was a Partial Drinking Bout or a Complete Drinking Bout.

The "Session Time" is how long the program will run. Program expects a Resolution of 1ms.

Compare the configurations below with the wiring chart supplied with all system orders. Edit input and output numbers if they are different for the system, or contact MED Associates for assistance.

Standard Input Configuration:

Liquid Head Entry = 1 (Level Mode)

Food Head Entry = 2 (Level Mode)

Standard Output Configuration:

House Light = 7

The sample program is as follows:

```
\ Copyright (C) 2008 MED Associates, All rights reserved.

\ Liquid & Food Weighing System.mpc
\
\ After the program starts it looks for a Head Entry into the Food and/or Liquid
\ Hopper.  When a Head Entry is detected the program times how long the Meal
\ and/or Drinking Session is and records how much food and/or liquid is taken
\ during the session.
\
\ The "Minimum Meal Weight" parameter is used to determine if the Eating Bout is
\ a Partial Meal or a Complete Meal.
\
\ The "End of Meal Criterion" is used to determine when a Meal has ended.  The
\ default value is two minutes.  This means that two minutes must pass with no
\ Head Entry into the Food Hopper before the Meal is considered over.  It is not
\ used to determine if the Meal was a Partial Meal or a Complete Meal.
\
\ The "Minimum Drink Weight" parameter is used to determine if the Drinking
\ Session is a Partial Drinking Bout or a Complete Drinking Bout.
\
\ The "End of Drinking Bout Criterion" is used to determine when a Drinking
\ Session has ended.  The default value is two minutes.  This means that two
\ minutes must pass with no Head Entry into the Liquid Hopper before the
\ Drinking Session is considered over.  It is not used to determine if the
\ Drinking session was a Partial Drinking Bout or a Complete Drinking Bout.
\
\ The "Session Time" is how long the program will run.
\
\ Program expects a Resolution of 1ms.


\ Inputs
^LiquidHeadEntry = 1  \ Level Mode
^FoodHeadEntry   = 2  \ Level Mode

\ Outputs
^HouseLight = 7


\ A() = Control Variables with Assigned Aliases as Defined
Var_Alias Minimum Meal Weight (grams)              = A(0)  \ Default = 0.1 grams
Var_Alias End of Meal Criterion (seconds)          = A(1)  \ Default = 120 seconds
Var_Alias Minimum Drink Weight (grams)             = A(2)  \ Default = 0.1 grams
Var_Alias End of Drinking Bout Criterion (seconds) = A(3)  \ Default = 120 seconds
Var_Alias Session Time (minutes)                   = A(4)  \ Default = 480 minutes

^MinMeal       = 0
^EndMeal       = 1
^MinDrink      = 2
^EndDrink      = 3
^Session       = 4
^EndMealTicks  = 5
^EndDrinkTicks = 6


\ List Data Variables Here
\  D = Total Liquid Drank (grams)
\  E = Total Food Eaten (grams)
\  F = Starting Food Total (Starting Weight) (grams)
\  G = Starting Liquid Total (Starting Weight) (grams)

\  M() = Meal Data
\  M(0)   = Total Number of Meals
\  M(1)   = Total Food Consumed in Meals (grams)
\  M(I)   = Length of Meal (ms)
\  M(I+1) = Food Consumed During Meal (grams)

\  N() = Drinking Bout Data
\  N(0)   = Total Number of Drinking Bouts
\  N(1)   = Total Liquid Consumed in Drinking Bouts (grams)
\  N(J)   = Length of Drinking Bout (ms)
\  N(J+1) = Liquid Consumed During Drinking Bout (grams)

\  P() = Partial Meal Data
\  P(0)   = Total Number of Partial Meals
\  P(1)   = Total Food Consumed in Partial Meals (grams)
\  P(K)   = Length of Partial Meal (ms)
\  P(K+1) = Food Consumed During Partial Meal (grams)
```

```
\  Q() = Partial Drinking Bout Data
\  Q(0)   = Total Number of Partial Drinking Bouts
\  Q(1)   = Total Liquid Consumed in Partial Drinking Bouts (grams)
\  Q(L)   = Length of Partial Drinking Bout (ms)
\  Q(L+1) = Liquid Consumed During Partial Drinking Bout (grams)

\  W = Ending Food Total (Ending Weight) (grams)
\  X = Ending Liquid Total (Ending Weight) (grams)


\ List Working Variables Here
\  B = Bout Food Eaten (grams)
\  C = Current Food Available (grams)
\  H = Current Liquid Available (grams)
\  I = Subscript for the Meal Array M
\  J = Subscript for the Liquid Bout Array N
\  K = Subscript for the Partial Meal Array P
\  L = Subscript for the Partial Liquid Bout Array Q
\  O = Bout Liquid Drank (grams)
\  R = Elapsed Session Timer
\  S = Meal Timer
\  T = Drinking Bout Timer


\ Z-Pulses Used in this Program
^StartMeal     = 1   \ Signal Start of a Meal
^StopMeal      = 2   \ Signal Stop of a Meal
^StartDrinking = 3   \ Signal Start of a Drinking Bout
^StopDrinking  = 4   \ Signal Stop of a Drinking Bout
^End           = 32  \ Signal End of Data Collection


DIM A = 6
DIM M = 1000
DIM N = 1000
DIM P = 1000
DIM Q = 1000


DISKFORMAT = 8.3
DISKCOLUMNS = 2


\************************************************
\      LIQUID & FOOD WEIGHING SYSTEM SCHEDULE
\  S1 - Set default values
\       Minimum Meal Weight              = 0.1 g
\       End of Meal Criterion            = 120 s
\       Minimum Drink Weight             = 0.1 g
\       End of Drinking Bout Criterion   = 120 s
\       Session Time                     = 480 min
\************************************************
S.S.1,
S1,
  0.01": SET A(^MinMeal)  = 0.1, A(^EndMeal) = 120, A(^MinDrink) = 0.1;
         SET A(^EndDrink) = 120, A(^Session) = 480;
         SET I = 2, J = 2, K = 2, L = 2;
         SET M(I) = -987.987, N(J) = -987.987;
         SET P(K) = -987.987, Q(L) = -987.987 ---> S2

S2,
  #START: CLEAR 1,60; ON ^HouseLight;
          SET A(^EndMealTicks)  = A(^EndMeal)  * 1";
          SET A(^EndDrinkTicks) = A(^EndDrink) * 1" ---> S3
  0.01": SHOW 2,Minimum Meal Weight (grams),A(^MinMeal),   3,End of Meal Criterion
(sec),A(^EndMeal);
         SHOW 4,Minimum Drink Weight (grams),A(^MinDrink), 5,End of Drinking Bout
Criterion (sec),A(^EndDrink);
         SHOW 6,Session Time (min),A(^Session) ---> SX

S3,
  1': SET A(^EndMealTicks)  = A(^EndMeal)  * 1";
      SET A(^EndDrinkTicks) = A(^EndDrink) * 1" ---> SX
```

```
\************************************************
\                  RECORD MEALS
\************************************************
S.S.2,
S1,
  #START: ~F := Weigh(MG, BOX+8);~;
          SHOW 11,Starting Food Weight,F ---> S2

S2,
  #R^FoodHeadEntry: Z^StartMeal ---> S3
  #Z^End: ~W := Weigh(MG, BOX+8);~ ---> S4

S3,
  #R^FoodHeadEntry: ---> S3
  A(^EndMealTicks)#T: ~W := Weigh(MG, BOX+8);~; SET B = C - W; Z^StopMeal;
                      IF B >= A(^MinMeal) [@Meal, @NoMeal]
                         @Meal: ADD M(0); SET M(I) = S, M(I+1) = B, M(I+2) = -987.987;
                                SET I = I + 2, M(1) = M(1) + B, E = E + B, C = W ---> S2
                         @NoMeal: ADD P(0); SET P(K) = S, P(K+1) = B, P(K+2) = -987.987;
                                  SET K = K + 2, P(1) = P(1) + B, E = E + B, C = W ---> S2
  #Z^End: ~W := Weigh(MG, BOX+8);~; SET B = C - W;
          IF B >= A(^MinMeal) [@Meal, @NoMeal]
             @Meal: ADD M(0); SET M(I) = S, M(I+1) = B, M(I+2) = -987.987;
                    SET I = I + 2, M(1) = M(1) + B, E = E + B, C = W ---> S4
             @NoMeal: ADD P(0); SET P(K) = S, P(K+1) = B, P(K+2) = -987.987;
                      SET K = K + 2, P(1) = P(1) + B, E = E + B, C = W ---> S4

S4,
  1': ---> SX


\************************************************
\                   MEAL TIMER
\************************************************
S.S.3,
S1,
  #Z^StartMeal: SET S = 0.001 ---> S2

S2,
  0.001": SET S = S + 0.001 ---> S2
  #Z^StopMeal ! #Z^End: ---> S1


\************************************************
\                 RECORD DRINKING
\************************************************
S.S.4,
S1,
  #START: ~G := Weigh(MG, BOX);~;
          SHOW 12,Starting Liquid Weight,G ---> S2
S2,
  #R^LiquidHeadEntry: Z^StartDrinking ---> S3
  #Z^End: ~X := Weigh(MG, BOX);~ ---> S4

S3,
  #R^LiquidHeadEntry: ---> S3
  A(^EndDrinkTicks)#T: ~X := Weigh(MG, BOX);~; SET O = H - X; Z^StopDrinking;
                       IF O >= A(^MinDrink) [@Drink, @NoDrink]
                          @Drink: ADD N(0); SET N(J) = T, N(J+1) = O, N(J+2) = -987.987;
                                  SET J = J + 2, N(1) = N(1) + O, D = D + O, H = X ---> S2
                          @NoDrink: ADD Q(0); SET Q(L) = T, Q(L+1) = O, Q(L+2) = -987.987;
                                    SET L = L + 2, Q(1) = Q(1) + O, D = D + O, H = X --->
S2
  #Z^End: ~X := Weigh(MG, BOX);~; SET O = H - X;
          IF O >= A(^MinDrink) [@Drink, @NoDrink]
             @Drink: ADD N(0); SET N(J) = T, N(J+1) = O, N(J+2) = -987.987;
                     SET J = J + 2, N(1) = N(1) + O, D = D + O, H = X ---> S4
             @NoDrink: ADD Q(0); SET Q(L) = T, Q(L+1) = O, Q(L+2) = -987.987;
                       SET L = L + 2, Q(1) = Q(1) + O, D = D + O, H = X ---> S4

S4,
  1': ---> SX
```

```
\*************************************************
\                DRINKING BOUT TIMER
\*************************************************
S.S.5,
S1,
  #Z^StartDrinking: SET T = 0.001 ---> S2

S2,
  0.001": SET T = T + 0.001 ---> S2
  #Z^StopDrinking ! #Z^End: ---> S1


\*************************************************
\                  UPDATE DISPLAY
\*************************************************
S.S.6,  \ Update Display
S1,
  #START: ---> S2

S2,
  1": SHOW 2,Total # of Meals,M(0),                  3,Total Food Eaten in Meals,M(1);
      SHOW 4,Total # of Partial Meals,P(0),          5,Total Food Eaten in Partial
Meals,P(1);
      SHOW 7,Total # of Drinking Bouts,N(0),         8,Total Liquid Taken in Drinking
Bouts,N(1);
      SHOW 9,Total # of Partial Drinking Bouts,Q(0), 10,Total Liquid Taken in Partial
Drinking Bouts,Q(1) ---> S2


\*************************************************
\                   TIME SESSION
\*************************************************
S.S.7,
S1,
  #START: ---> S2

S2,
  1": ADD R; SHOW 1,Session Time (min),R/60;
      IF R/60 >= A(^Session) [@End, @Continue]
         @End: OFF ^HouseLight; Z^End ---> S3
         @Continue: ---> S2

S3,     \ Allow time for screen update before closing the session and saving data.
  5": ---> STOPABORTFLUSH


\*************************************************
\                UPDATE FOOD WEIGHT
\*************************************************
S.S.8,  \ Get the Starting Weight.
        \
        \ Update the Starting Weight every 1 minute
        \ when the Animal is not eating.  This removes
        \ food that is lost for reasons other than the
        \ Animal eating.
S1,
  0.001": ~C := Weigh(MG, BOX+8);~; ---> S2

S2,
  #R^FoodHeadEntry: ---> S3
  1': ~C := Weigh(MG, BOX+8);~; ---> S2
  #Z^End: ---> S4

S3,
  #R^FoodHeadEntry: ---> S3
  A(^EndMealTicks)#T: ---> S2
  #Z^End: ---> S4

S4,
  1': ---> SX
```

```
\************************************************
\               UPDATE LIQUID WEIGHT
\************************************************
S.S.9,  \ Get the Starting Weight.
        \
        \ Update the Starting Weight every 1 minute
        \ when the Animal is not drinking.  This
        \ removes liquid that is lost for reasons
        \ other than the Animal drinking.
S1,
  0.01": ~H := Weigh(MG, BOX);~; ---> S2

S2,
  #R^LiquidHeadEntry: ---> S3
  1': ~H := Weigh(MG, BOX);~; ---> S2
  #Z^End: ---> S4

S3,
  #R^LiquidHeadEntry: ---> S3
  A(^EndDrinkTicks)#T: ---> S2
  #Z^End: ---> S4

S4,
  1': ---> SX
```

APPENDIX

Driver and Software Installation

If not already done, install the MED-PC software and the DIG-704 card per the instructions included with MED-PC. Do **NOT** install the DIG-744 PCI Data Acquisition card (PCI-6023E) at this time. You will be instructed to do so at a later time.

Insert the Intake Monitor CD into the CD-ROM drive. The screen shown in Figure A.1 will appear. Click on **Install Weighing System**. The screen shown in Figure A.2 will appear.

Figure A.1 – Weighing System Setup



Click **Install** to begin the driver and software installation. Follow the steps to complete the National Instruments installation and when installation is completed the screen shown in Figure A.3 will appear.

Figure A.2 – Begin Driver and Software Installation



The software and driver installation is now complete. Click **Finish** to close this window. Shut down the computer and remove power. Install the DIG-744 PCI Data Acquisition Card according to the instructions included with the computer for installing PCI cards.

Figure A.3 – Installation Complete